Proof that A356635 consists of all numbers divisible by at least one of **7, 8, 9, 10, 12, 15, 22, 33, 39, 52, 55, 68, 102, 114, 138.**

Robert Israel

2 Sep 2022

We consider all positive integer solutions of the equation $1=\frac{1}{a}+\frac{1}{b}+\frac{1}{c}+\frac{1}{d}+\frac{1}{e}+\frac{1}{f}+\frac{1}{g}$. Without loss of generality, we may assume $a \le b \le c \le d \le e \le f \le g$. In particular, $\frac{7}{a} \ge \frac{1}{a}+\frac{1}{b}+\frac{1}{c}+\frac{1}{d}+\frac{1}{e}+\frac{1}{f}+\frac{1}{g}=1$ so $a \le 7$. Similarly, given a, $\frac{6}{b} \ge \frac{1}{b}+\frac{1}{c}+\frac{1}{d}+\frac{1}{e}+\frac{1}{f}+\frac{1}{g}=1-\frac{1}{a}$, so $b \le \frac{6}{1-\frac{1}{a}}$. On the other hand, $\frac{1}{a}+\frac{1}{b}<1$ so $b>\frac{1}{1-\frac{1}{a}}$. Similarly we can get upper and lower bounds for each of the variables in terms of the previous ones, and enumerate all solutions $[a, b, c, d, e, f, g]$ using the following Maple program.

```
> R:= NULL:
  for a from 2 to 7 do
    x:= 1-1/a;
    for b from max(a,1+floor(1/x)) to floor(6/x) do
       y:= x-1/b;
       for c from max(b,1+floor(1/y)) to floor(5/y) do
         z:= y-1/c;
         for d from max(c,1+floor(1/z)) to floor(4/z) do
           w:= z-1/d;
           for e from max(d,1+floor(1/w)) to floor(3/w) do
             v:= w-1/e;
             for f from max(e,1+floor(1/v)) to floor(2/v) do
               g:= 1/(v-1/f); if g::integer and g >= f then R:= R,
  [a,b,c,d,e,f,g] fi
  od od od od od od:
```

$nops([R])$

$$294314 \tag{1}$$

It turns out there are 294314 solutions with $a \le b \le c \le d \le e \le f \le g$. Each corresponds to a possible pattern $k=\frac{k}{a}+\frac{k}{b}+\frac{k}{c}+\frac{k}{d}+\frac{k}{e}+\frac{k}{f}+\frac{k}{g}$ for writing k as the sum of 7 of its divisors which works if and only if k is divisible by each of a, b, c, d, e, f, g and thus by $lcm(a, b, c, d, e, f, g)$. We find all these lcm's.

```
> Lcms:= map( ilcm, {R}):
```

Now we find a minimal set such that all these lcm's are divisible by a member of the set.

```
> S:= Lcms: S1:= {}:
  while S <> {} do
    s:= min(S); S1:= S1 union {s};
    S:= remove(t -> t mod s = 0, S);
  od:
```

Here are the results:

```
> S1;
```

$$\{7, 8, 9, 10, 12, 15, 22, 33, 39, 52, 55, 68, 102, 114, 138\} \qquad (2)$$

Thus we conclude that k is a member of A356635 if and only if it is divisible by at least one of the numbers 7, 8, 9, 10, 12, 15, 22, 33, 39, 52, 55, 68, 102, 114, 138.

Let us find solutions corresponding to each of these.

```
> for m in S1 do
    E:= select(t -> ilcm(t) = m, [R])[1];
    printf("if k is divisible by %d, then k = k/%d + k/%d + k/%d +
  k/%d + k/%d + k/%d + k/%d\n", m, seq(E[i],i=1..7))
  od:
```

```
if k is divisible by 7, then k = k/7 + k/7 + k/7 + k/7 + k/7 + k/7 +
k/7
if k is divisible by 8, then k = k/4 + k/8 + k/8 + k/8 + k/8 + k/8 +
k/8
if k is divisible by 9, then k = k/3 + k/9 + k/9 + k/9 + k/9 + k/9 +
k/9
if k is divisible by 10, then k = k/5 + k/5 + k/5 + k/10 + k/10 +
k/10 + k/10
if k is divisible by 12, then k = k/2 + k/12 + k/12 + k/12 + k/12 +
k/12 + k/12
if k is divisible by 15, then k = k/3 + k/3 + k/15 + k/15 + k/15 +
k/15 + k/15
if k is divisible by 22, then k = k/2 + k/11 + k/11 + k/11 + k/11 +
k/11 + k/22
if k is divisible by 33, then k = k/3 + k/3 + k/11 + k/11 + k/11 +
k/33 + k/33
if k is divisible by 39, then k = k/3 + k/3 + k/13 + k/13 + k/13 +
k/13 + k/39
if k is divisible by 52, then k = k/2 + k/4 + k/13 + k/13 + k/26 +
k/26 + k/52
if k is divisible by 55, then k = k/5 + k/5 + k/5 + k/5 + k/11 + k/11
+ k/55
if k is divisible by 68, then k = k/2 + k/4 + k/17 + k/17 + k/17 +
k/17 + k/68
if k is divisible by 102, then k = k/2 + k/3 + k/17 + k/17 + k/34 +
k/102 + k/102
if k is divisible by 114, then k = k/2 + k/3 + k/19 + k/19 + k/38 +
k/38 + k/114
if k is divisible by 138, then k = k/2 + k/3 + k/23 + k/23 + k/23 +
k/46 + k/69
```